nCells 6-param double logistic.nb

This *Mathematica* 7 notebook presents, generates data from and retrieves the parameters of a 6-parameter continuous deterministic model of microbial growth and mortality, nCells6[t]. It is constructed from an underlying 3-parameter logistic model of cell's division probability, pDiv6[t], and a 3-parameter logistic model of cell's mortality, pMort6[t] having different asymptotes, pdAsym and pmAsym, steepness parameters, kd and km, and characteristic times, tcd and tcm, respectively. As a test, growth/morality data are first generated with this model and then the NonlinearModelFit function is used to fit them and retrieve the values of the 6 original generation parameters. The generated data and the fitted model are then plotted together for visual comparison. Also plotted are the corresponding probability functions pDiv6[t] and pMort6[t]. This is followed by the generation with the same model of some noisy data having a random scatter. The NonlinearModelFit function is used again in an attempt to recover the model's 6 parameters. The generated noisy data and the fitted curve are then plotted together for visual comparison and so are the new probability functions pDiv6[t] and pMort6[t]. The generated noisy data can be replaced by pasted experimental data and the procedure used to estimate the organism's growth/mortality parameters and the corresponding probability functions pDiv6[t] and pMort6[t] under the pertinent conditions.

Programmed by Mark D. Normand based on models developed and elaborated by Micha Peleg, Joseph Horowitz, Murray Eisenberg and Maria G. Corradini. Last modified: November 11, 2009

Clear all variables used in this notebook.

```
In[643]:= Clear[dt, maxNoise, n0, nCells6, nCells6Pts, nCells6PtsRand, nlr6Results, nlr6ResultsRand,
   nParams, nPts6, nPts6Rand, pDiv6, pMort6, pdAsym, pmAsym, kd, km, tcd, tcm, tSt, tEn,
   plot6pDivpMortExact, plot6pDivpMortGen, plot6pDivpMortRand, plot6GenExact, plot6FitExact,
   plot6FitGenExact, plot6GenRand, plot6FitGenRand, plot6FitRand, yDivMortMax, yMax, yMaxRand]
```

The pDiv6[t] function defines the probability of cell division at time t. The parameters of pDiv6[t] are pdAsym, kd and tcd.

```
In[644]:= pDiv6[t_] := pdAsym / (1 + Exp[kd * (tcd - t)])
```

The pMort6[t] function defines the probability of cell mortality at time t. The parameters of pMort6[t] are pmAsym, km and tcm.

```
In[645]:= pMort6[t_] := pmAsym / (1 + Exp[km * (tcm - t)])
```

n0 is the initial number of cells alive at time t=0.

```
In[646]:= n0 = 1000.;
```

Define the nCells6Integrate[t] function that combines the pDiv6[t] and pMort6[t] functions to give the number of cells alive at time t.

```
In[647]:= nCells6Integrate[t_] :=
   n0 * Exp[Subtract @@ (Integrate[pDiv6[tt] - pMort6[tt], tt] /. tt → {t, 0})]
```

If both pDiv6[t] and pMort6[t] can be integrated anlytically, calling nCells6Integrate[t] gives an explicit formula that does NOT include Integrate.

In[648]:= `nCells6Integrate[t]`

Out[648]= $1000.\ e^{\text{pdAsym}\ t-\text{pmAsym}\ t-\frac{\text{pdAsym Log}\left[1+e^{\text{kd tcd}}\right]}{\text{kd}}+\frac{\text{pdAsym Log}\left[1+e^{-\text{kd t}+\text{kd tcd}}\right]}{\text{kd}}+\frac{\text{pmAsym Log}\left[1+e^{\text{km tcm}}\right]}{\text{km}}-\frac{\text{pmAsym Log}\left[1+e^{-\text{km t}+\text{km tcm}}\right]}{\text{km}}}$

Now define the nCells6[t] function to be the explicit formula shown above WHICH DOES NOT include Integrate.

In[649]:= `nCells6[t_] = nCells6Integrate[t]`

Out[649]= $1000.\ e^{\text{pdAsym}\ t-\text{pmAsym}\ t-\frac{\text{pdAsym Log}\left[1+e^{\text{kd tcd}}\right]}{\text{kd}}+\frac{\text{pdAsym Log}\left[1+e^{-\text{kd t}+\text{kd tcd}}\right]}{\text{kd}}+\frac{\text{pmAsym Log}\left[1+e^{\text{km tcm}}\right]}{\text{km}}-\frac{\text{pmAsym Log}\left[1+e^{-\text{km t}+\text{km tcm}}\right]}{\text{km}}}$

nParams is the number of different parameters in the nCells6[t] model. Assign initial values to all 6 parameter variables. These values will be used to generate a list of {t, nCells6[t]} data values.

In[650]:= `nParams = 6; pdAsym = 0.25; pmAsym = 0.2; kd = 0.5; km = 0.4; tcd = 20.; tcm = 10.;`

tSt and tEn are the starting and ending times over which to generate and plot the data. dt is the delta-t time increment between data points.

In[651]:= `tSt = 0.; tEn = 40.; dt = 4.;`

minOffset is used in the plots to force plotting the axis tick mark label at 0.

In[652]:= `minOffset = .001;`

Assign the maximum y-axis limit of the pDiv6 and pMort6 plots to yDivMortMax.

In[653]:= `pDiv6[tEn]`

Out[653]= 0.249989

In[654]:= `pMort6[tEn]`

Out[654]= 0.199999

In[655]:= `yDivMortMax = 0.3;`

Plot the pDiv6[t] and pMort6[t] functions together on the same plot using the currently assigned values of the 6 parameters used to generate the exact data.

In[656]:=
```
plot6pDivpMortGen = Plot[{pDiv6[t], pMort6[t]}, {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True,
  PlotLabel → ToString[nParams] <> "-par. data generation pDiv6[t] (blue) & pMort6[t] (red)",
  FrameLabel → {"time", "pDiv6[t] & pMort6[t]"}]
```

Out[656]=



Generate, fit and plot some exact data (nCells6Pts) using the 6-parameter nCells6[t] function.

nCells6Pts is a generated list of {t, nCells6[t]} ordered data pairs to be used in curve fitting with nonlinear regression and plotting.

In[657]:=
```
nCells6Pts = Table[{t, nCells6[t]}, {t, tSt, tEn, dt}]
```

Out[657]=
```
{{0., 1000.}, {4., 966.38}, {8., 839.239}, {12., 566.985}, {16., 310.087}, {20., 191.389},
 {24., 177.403}, {28., 205.513}, {32., 249.129}, {36., 303.979}, {40., 371.232}}
```

nPts6 is the number of generated data points in the nCells6Pts list.

In[658]:=
```
nPts6 = Length[nCells6Pts]
```

Out[658]=
```
11
```

Assign the maximum y-axis value to be used in the exact data plots to yMax.

In[659]:=
```
Max[nCells6Pts[[All, 2]]]
```

Out[659]=
```
1000.
```

In[660]:=
```
yMax = 1200.;
```

Clear all the variables to be used by NonlinearModelFit.

In[661]:=
```
Clear[nlr6Results, pdAsym, pmAsym, kd, km, tcd, tcm]
```

Call the NonlinearModelFit function which adjusts the values of the 6 parameters to give the best fit of the nCells6[t] function to the nCells6Pts data and return a report in nlr6Results. All 6 parameters are unconstrained.

In[662]:=
```
nlr6Results = NonlinearModelFit[nCells6Pts, nCells6[t],
  {{pdAsym, 0.20}, {kd, 0.4}, {tcd, 10}, {pmAsym, 0.16}, {km, 0.3}, {tcm, 15}},
  t, MaxIterations → 500]
```

Out[662]= FittedModel$\left[\frac{1000.\, e^{\ll 1 \gg} (1 + \ll 1 \gg)^{\ll 19 \gg}}{\left(1 + e^{\ll 1 \gg}\right)^{\ll 19 \gg}}\right]$

Show the best fitted model, the best fit parameters, the parameter table, the R-squared coefficient of determination, the value of R-squared adjusted for the number of model parameters, the Akaike Information Criterion and the Baysian Information Criterion from the nlr6Results report.

In[663]:=
```
nlr6Results[{"BestFit", "BestFitParameters",
  "ParameterTable", "RSquared", "AdjustedRSquared", "AIC", "BIC"}]
```

Out[663]= $\left\{\frac{1000.\, e^{-2.99095+0.05\, t} \left(1 + e^{10.-0.5\, t}\right)^{0.5}}{\left(1 + e^{4.-0.4\, t}\right)^{0.5}},\right.$

{pdAsym → 0.25, kd → 0.5, tcd → 20., pmAsym → 0.2, km → 0.4, tcm → 10.},

	Estimate	Standard Error	t Statistic	P-Value
pdAsym	0.25	3.96451×10^{-15}	6.30595×10^{13}	1.90349×10^{-68}
kd	0.5	1.19962×10^{-14}	4.16798×10^{13}	1.50895×10^{-67}
tcd	20.	7.42712×10^{-14}	2.69283×10^{14}	1.34047×10^{-71}
pmAsym	0.2	3.85485×10^{-15}	5.18827×10^{13}	5.04883×10^{-68}
km	0.4	3.84328×10^{-15}	1.04078×10^{14}	1.55422×10^{-69}
tcm	10.	9.3492×10^{-14}	1.06961×10^{14}	1.35574×10^{-69}

, 1., 1., -583.521, -580.736}

Extract the fitted values of the 6 parameters from the nlr6Results report and assign them to the 6 parameter variables.

In[664]:=
```
nlr6Results["BestFitParameters"]
```

Out[664]= {pdAsym → 0.25, kd → 0.5, tcd → 20., pmAsym → 0.2, km → 0.4, tcm → 10.}

In[665]:=
```
pdAsym = nlr6Results["BestFitParameters"][[1, 2]]
```

Out[665]= 0.25

In[666]:=
```
kd = nlr6Results["BestFitParameters"][[2, 2]]
```

Out[666]= 0.5

In[667]:=
```
tcd = nlr6Results["BestFitParameters"][[3, 2]]
```

Out[667]= 20.

In[668]:=
```
pmAsym = nlr6Results["BestFitParameters"][[4, 2]]
```

Out[668]= 0.2

In[669]:=
```
km = nlr6Results["BestFitParameters"][[5, 2]]
```

Out[669]= 0.4

In[670]:=
```
tcm = nlr6Results["BestFitParameters"][[6, 2]]
```

Out[670]= 10.

plot6GenExact is a ListPlot of the nCells6Pts exact data generated using the 6-parameter nCells6[t] function.

In[671]:=
```
plot6GenExact =
 ListPlot[nCells6Pts, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMax}},
  Frame → True, Axes -> False, PlotStyle → {PointSize → Medium, Blue},
  PlotLabel → ToString[nPts6] <> " nCells6Pts data at dt=" <> ToString[dt] <> " from " <>
    ToString[nParams] <> "-par. model", FrameLabel → {"time", "nCells6Pts data", "", ""}]
```

Out[671]=



plot6FitExact is a Plot of the nCells6[t] function using the currently assigned values of the 6 parameters fitted to the nCells6Pts exact data.

In[672]:=
```
plot6FitExact =
 Plot[nCells6[t], {t, tSt, tEn}, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMax}},
  Frame → True, Axes -> False, PlotStyle → {Blue},
  PlotLabel → ToString[nParams] <> "-par. nCells6[t] curve fitted to nCells6Pts data",
  FrameLabel → {"time", "nCells6[t]", "", ""}]
```

Out[672]=



Show the 6-parameter nCells6[t] function curve and the nCells6Pts exact data values together on the same plot.

In[673]:=
```
plot6FitGenExact = Show[plot6FitExact, plot6GenExact,
  PlotLabel → ToString[nPts6] <> " nCells6Pts data at dt=" <>
    ToString[dt] <> " fitted with " <> ToString[nParams] <> "-par. nCells6[t]",
  FrameLabel → {"time", "nCells6Pts data & nCells6[t]", "", ""}]
```

Out[673]=



Plot the pDiv6[t] and pMort6[t] functions together on the same plot using the currently assigned values of the 6 parameters fitted to the nCells6Pts exact data.

In[674]:=
```
plot6pDivpMortExact = Plot[{pDiv6[t], pMort6[t]}, {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True,
  PlotLabel → ToString[nParams] <> "-par. nCells6Pts fit pDiv6[t] (blue) & pMort6[t] (red)",
  FrameLabel → {"time", "pDiv6[t] & pMort6[t]"}]
```

Out[674]=



Generate, fit and plot some noisy data (nCells6PtsRand) using the 6-parameter nCells6[t] function.

WARNING: Executing the following cell makes *Mathematica* always generate the same repeatable sequence of pseudo-random values from the specified seed value. Run without executing the next cell to restore pseudo-randomness or change the assigned seed value to get a different repeatable pseudo-random sequence.

In[675]:=
```
seed = 0; SeedRandom[seed]
```

maxNoise is the maximum allowed noise value. It must be less than or equal to the initial number of cells, n0.

In[676]:=
```
maxNoise = 50.;
```

nCells6PtsRand is the nCells6Pts list of ordered data pairs in which random noise has been added to the nCells6[t] coordinate values.

If you wish to fit and plot experimental data, assign them to nCells6PtsRand in the same form as shown below.

In[677]:=
```
nCells6PtsRand = Table[{t, nCells6[t] + RandomReal[{-maxNoise, maxNoise}]}, {t, tSt, tEn, dt}]
```

Out[677]=
```
{{0., 1015.25}, {4., 979.687}, {8., 857.52}, {12., 573.62}, {16., 353.607}, {20., 239.007},
 {24., 151.249}, {28., 219.269}, {32., 209.239}, {36., 318.532}, {40., 337.184}}
```

nPts6Rand is the number of generated data points in the nCells6PtsRand list.

In[678]:=
```
nPts6Rand = Length[nCells6PtsRand]
```

Out[678]=
```
11
```

Assign the maximum y-axis value to be used in the noisy data plots to yMaxRand.

In[679]:=
```
Max[nCells6PtsRand[[All, 2]]]
```

Out[679]=
```
1015.25
```

In[680]:=
```
yMaxRand = 1200.;
```

Clear all the variables to be used by NonlinearModelFit.

In[681]:=
```
Clear[nlr6ResultsRand, pdAsym, pmAsym, kd, km, tcd, tcm]
```

Call the NonlinearModelFit function which adjusts the values of the 6 parameters to give the best fit of the nCells6[t] function to the nCells6PtsRand noisy data and return a report in nlr6ResultsRand. All 6 parameters are unconstrained.

Note that since the data are dependent on random numbers, the nonlinear fit may be very sensitive to the initial guesses of the values of the 6 parameters.

In[682]:=
```
nlr6ResultsRand = NonlinearModelFit[nCells6PtsRand, nCells6[t],
  {{pdAsym, 0.20}, {kd, 0.4}, {tcd, 10.}, {pmAsym, 0.16}, {km, 0.3}, {tcm, 15.}},
  t, MaxIterations → 500]
```

NonlinearModelFit::cvmit :
Failed to converge to the requested accuracy or precision within 500 iterations. ≫

Out[682]= FittedModel[$\frac{1000.\, e^{\ll 1 \gg} (1 + \ll 1 \gg)^{\ll 19 \gg}}{(1 + e^{\ll 1 \gg})^{\ll 19 \gg}}$]

Show the best fitted model, the best fit parameters, the parameter table, the R-squared coefficient of determination, the value of R-squared adjusted for the number of model parameters, the Akaike Information Criterion and the Baysian Information Criterion from the nlr6Results report.

In[683]:=
```
nlr6ResultsRand[{"BestFit", "BestFitParameters",
  "ParameterTable", "RSquared", "AdjustedRSquared", "AIC", "BIC"}]
```

Out[683]=

$$\left\{\frac{1000.\ e^{-3.48757+0.060834\,t}\left(1+e^{3.05008-0.236451\,t}\right)^{24.301}}{\left(1+e^{3.04597-0.245007\,t}\right)^{23.2041}},\right.$$

{pdAsym → 5.746, kd → 0.236451, tcd → 12.8994, pmAsym → 5.68517, km → 0.245007, tcm → 12.4322},

	Estimate	Standard Error	t Statistic	P-Value
pdAsym	5.746	4679.57	0.00122789	0.999068
kd	0.236451	3.25097	0.0727324	0.944839
tcd	12.8994	199.616	0.0646213	0.95098
pmAsym	5.68517	4679.58	0.00121489	0.999078
km	0.245007	3.75733	0.0652077	0.950536
tcm	12.4322	183.013	0.0679305	0.948474

, 0.998657, 0.997046, 112.056, 114.841}

Extract the fitted values of the 6 parameters from the nlr6ResultsRand report and assign them to the 6 parameter variables.

In[684]:=
```
nlr6ResultsRand["BestFitParameters"]
```

Out[684]= {pdAsym → 5.746, kd → 0.236451, tcd → 12.8994, pmAsym → 5.68517, km → 0.245007, tcm → 12.4322}

In[685]:=
```
pdAsym = nlr6ResultsRand["BestFitParameters"][[1, 2]]
```

Out[685]= 5.746

In[686]:=
```
kd = nlr6ResultsRand["BestFitParameters"][[2, 2]]
```

Out[686]= 0.236451

In[687]:=
```
tcd = nlr6ResultsRand["BestFitParameters"][[3, 2]]
```

Out[687]= 12.8994

In[688]:=
```
pmAsym = nlr6ResultsRand["BestFitParameters"][[4, 2]]
```

Out[688]= 5.68517

In[689]:=
```
km = nlr6ResultsRand["BestFitParameters"][[5, 2]]
```

Out[689]= 0.245007

In[690]:=
```
tcm = nlr6ResultsRand["BestFitParameters"][[6, 2]]
```

Out[690]= 12.4322

plot6GenRand is a ListPlot of the nCells6PtsRand noisy data generated using the 6-parameter nCells6[t] function.

In[691]:=
```
plot6GenRand =
 ListPlot[nCells6PtsRand, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMaxRand}},
  Frame → True, Axes -> False, PlotStyle → {PointSize → Medium, Red},
  PlotLabel → ToString[nPts6Rand] <> " nCells6PtsRand data at dt=" <>
    ToString[dt] <> " from " <> ToString[nParams] <> "-par. model",
  FrameLabel → {"time", "nCells6PtsRand data", "", ""}]
```

Out[691]=



plot6FitRand is a Plot of the nCells6[t] function using the currently assigned values of the 6 parameters fitted to the nCells6PtsRand noisy data..

In[692]:=
```
plot6FitRand = Plot[nCells6[t], {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMaxRand}},
  Frame → True, Axes -> False, PlotStyle → {Red},
  PlotLabel → ToString[nParams] <> "-par. nCells6[t] curve fitted to nCells6PtsRand data",
  FrameLabel → {"time", "nCells6[t]", "", ""}]
```

Out[692]=



Show the fitted 6-parameter nCells6[t] function curve and the nCells6PtsRand noisy data values together on the same plot.

In[693]:=
```
plot6FitGenRand = Show[plot6FitRand, plot6GenRand,
  PlotLabel → ToString[nPts6Rand] <> " nCells6PtsRand data at dt=" <>
    ToString[dt] <> " fitted with " <> ToString[nParams] <> "-par. nCells6[t]",
  FrameLabel → {"time", "nCells6PtsRand data & nCells6[t]", "", ""}]
```

Out[693]=

11 nCells6PtsRand data at dt=4. fitted with 6−par. nCells6[t]



Plot the pDiv6[t] and pMort6[t] functions together on the same plot using the currently assigned values of the 6 parameters fitted to the nCells6PtsRand noisy data.

In[694]:=
```
plot6pDivpMortRand = Plot[{pDiv6[t], pMort6[t]}, {t, tSt, tEn}, PlotRange →
    {{tSt - minOffset, tEn}, {0. - minOffset, 6. + minOffset}}, Frame → True, PlotLabel →
   ToString[nParams] <> "-par. nCells6PtsRand fit pDiv6[t] (blue) & pMort6[t] (red)",
  FrameLabel → {"time", "pDiv6[t] & pMort6[t]"}]
```

Out[694]=

6−par. nCells6PtsRand fit pDiv6[t] (blue) & pMort6[t] (red)